

Paper 129-31

## Incorporating PROC FORMAT, PROC REPORT, and ODS Style Definitions to Track Sources of Records with Poor Data

Thaer Baroud, MHSA, MA, John Senner, PhD, Paul Johnson, MSPH, Little Rock, Arkansas

### ABSTRACT

Data quality is essential to proper planning, evaluation, and research in nearly all industries. This paper describes how PROC FORMAT and PROC REPORT can be employed to generate a color-coded report that visibly displays sources of data values outside acceptable ranges.

Using birth certificate data collected and submitted by hospitals, this report examines multiple data fields against specified criteria and accordingly assigns different colors. By glancing at the report, the reader can quickly identify hospitals with reporting problems in several areas or if there is a particular item that is a problem for many hospitals.

### INTRODUCTION

The excellence and completeness of birth certificate data is vital in public health research, planning and services. State and Federal Maternal and Child Health (MCH) programs utilize birth certificate data to target mothers from particular racial/ethnic and age groups with low utilization of prenatal care or to investigate maternal characteristics associated with low birth weight, prematurity, infant mortality and other unfavorable birth outcomes.

Arkansas State laws require birth certificates to be completed for all births, and Federal law mandates national collection and publication of births and other vital statistics data<sup>1</sup>. The National Vital Statistics System (NVSS), the Federal compilation of this data, is the result of cooperation between CDC's National Center for Health Statistics (NCHS) and state agencies that brings about a wealth of statistical information from birth certificates.

NCHS, every year, sets up standards of acceptable proportions of poor data fields reported as not classifiable (i.e., outside predetermined ranges), unknown or blank in birth certificate data. Since the vast majority of birth certificates are collected and submitted by hospitals, the Center for Health Statistics at the Arkansas Division of Health needed a report that can easily identify fields with high proportions of invalid data by hospital of birth. Such a report is important for administrators of vital records and field representatives who can contact these hospitals to provide educational seminars to nurses and clerks on appropriate methods in collecting and reporting birth data.

With the introduction of ODS, we wanted to take advantage of its strong reporting capabilities and the ability to create appealing tables, to craft a report that can be used to identify hospitals that contribute to high quantities of poor data.

### THE ASSIGNMENT

Using birth data submitted by 54 states and territories, NCHS produces an annual state-level report entitled: "NATALITY NOT CLASSIFIABLE AND UNKNOWN" and makes it available to Centers for Health Statistics and Vital Records, or any other individual state agencies that collect, edit, and publish vital statistics data.

In 2003, NCHS selected 40 data fields to report on. The report includes the acceptable percentage of records (standard) that are not classifiable, blank and unknown. Standards are computed using criteria of multiplying the nationwide median for 1998 for a particular item, as a percentage of records received by NCHS, by a multiplier. The multiplier was 1.5 for 2003, which is 0.5 less than that of 2002. The following is an example of how the standard was computed for prenatal visits in different years.

Prenatal Visits	1998 Median (%)	2000 Standard = 3X the Median for 1998	2001 Standard = 2.5X the Median for 1998	2002 Standard = 2X the Median for 1998	2003 Standard = 1.5X the Median for 1998
	1.99	5.97	4.98	3.98	2.99

Since most births occur in hospitals and freestanding birth facilities, which are required to submit birth certificate data to the state health department, we wanted to generate an Arkansas-specific report by hospital of birth using the same standards supplied by NCHS. We also wanted to identify fields in which Arkansas exceeds the NCHS standard for that particular year, as well as to distinguish hospitals that exceed that standard. One senior analyst suggested that the report also highlights fields that may potentially exceed the standard for both the state and individual hospitals.

For purposes of this paper, we have selected 8 birth certificate items that are looked at more often in analyzing preterm birth\*. These are: clinical estimate of gestation (*gest*), day last normal menses began (*menda*), month last menses began (*menmo*), year last menses began (*menyr*), month prenatal care began (*care*), named month prenatal care began (*nammo*), prenatal visits (*visits*), and weight gain during pregnancy (*wtgain*).

## BEFORE ODS

Before we learned about ODS, we generated a similar report but used Microsoft Excel to present the tables. We categorized the selected variables, summarized the data using PROC SUMMARY and exported the data table to Excel. In Excel, we customized some Visual Basic for Applications (VBA) scripts generated by running user-defined Excel macros to color-shade cells where values exceeded the standards. With the introduction of ODS and the powerful PROC REPORT, we thought of generating the report “by clicking only one button” entirely in a SAS environment.

## THE SAS PROGRAM

### GETTING STARTED

```

OPTIONS PAGENO = 1 CENTER PAPER SIZE = LEGAL ORIENTATION = LANDSCAPE; ❶
LIBNAME birth 'path-to-server/birth database';
%LET data = birth.birth2003; ❷
%LET path = C:\SUGI31; ❸

```

To be able to generate a report with all your summarized variables (numbers and percents) lined up in one page, use PAPER SIZE = LEGAL and ORIENTATION = LANDSCAPE options in the OPTIONS statement ❶. The %LET macro-program statement ❷, the simplest way to create and assign a value to a macro variable, is used to assign a value to the permanent birth data file and %LET statement ❸ defines the path where you want to store your generated report (in a PDF file format, as discussed later).

```

DATA temp (KEEP = ostate bhosp_ bloc gest gestl menda mendal menmo
              menmol menyr menyrl care carel nammo nammol
              visits visitsl wtgain_ wtgainl rec_err); ❶
  SET &data (PW = *****);
  IF bloc NOT IN ('1', '2') THEN bhosp_ = 'Non-Hosp'; ❷

  IF NOT ('02' LE gest LE '44') THEN gestl = 1;
  ELSE gestl = 0;
  IF NOT ('01' LE menda LE '31') THEN mendal = 1;
  ELSE mendal = 0;
  IF NOT ('01' LE menmo LE '12') THEN menmol = 1;
  ELSE menmol = 0;
  IF menyr IN ('9999', ' ', '0000') THEN menyrl = 1;
  ELSE menyrl = 0;
  IF NOT ('00' LE care LE '09') THEN carel = 1;
  ELSE carel = 0;
  /*CODE ONLY IF "MONTH PRENATAL CARE BEGAN" IS BLANK*/ ❸
  IF care EQ ' ' and NOT ('01' LE nammo LE '12') THEN nammol = 1;
  ELSE nammol = 0;
  IF NOT ('00' LE visits LE '97') THEN visitsl = 1;
  ELSE visitsl = 0;
  IF wtgain_ IN (999,.) OR wtgain_ GT 200 THEN wtgainl = 1;
  ELSE wtgainl = 0;
  IF gestl + mendal + menmol + menyrl +
    carel + nammol + visitsl + wtgainl gt 0 THEN rec_err = 1;
  ELSE rec_err = 0;
RUN;

```

DATA step ❶ creates a temporary analysis data set and the KEEP =, a data step option, reduces the number of variables to those that you will be using throughout the program in the work dataset. The IF-THEN statement ❷ recodes birth hospital numbers for non-hospital births to ‘Non-Hosp’. Evaluating this type of birth certificates uncovers

\* Arkansas has been experiencing a decline in preterm birth rate in the last five years. The accuracy and completeness of pertinent data is essential to analyzing the significance of the trend.

the quality of mostly home-birth certificate data. IF-THEN/ELSE group of arguments ❸ creates new analysis variables from original birth variables of interest. In these newly created variables, valid data entries (within an acceptable range) will be assigned 0s, while not-classifiables, unknowns, blanks, and other invalid entries will be assigned 1s. Note that the last IF-THEN/ELSE argument creates a new variable (*rec\_err*), which is coded 1 if a birth record contains at least one invalid entry in the selected fields. Clean birth records are coded 0.

#### STEP 1: SUMMARIZE YOUR DATA USING THE SUMMARY PROCEDURE

```
PROC SUMMARY DATA = temp; ❶
  WHERE ostate EQ '004'; ❷
  CLASS bhosp_; ❸
  VAR gestl -- rec_err; ❹
  OUTPUT OUT = summary (DROP = _TYPE_ RENAME = (_FREQ_ = Birth))
  SUM = MEAN = / AUTONAME; ❺
RUN;
```

As the name implies, the “good old” SUMMARY procedure can be used to summarize data by computing counts and percentages. PROC SUMMARY statement ❶ invokes the procedure and the WHERE statement ❷ requests that the summarization be limited to births that occurred in Arkansas, as we are interested in in-state birth hospitals only. Since we need to summarize the data at the hospital level, variable *bhosp\_* (birth hospital code) is inserted as a class-variable in the CLASS statement. The range of variables listed in the VAR statement ❹ represents the 8 selected birth certificate items for the report together with the newly created *rec\_err* (true if record contains at least one invalid entry) - a total of 9 variables.

Request the output of the SUMMARY procedure be routed to a work data set (*summary*) by using the OUTPUT OUT = statement ❺. When you output your summarized results into a SAS data set, PROC SUMMARY, by default, generates two variables: *\_TYPE\_*, and *\_FREQ\_*. Values in variable *\_TYPE\_* correspond to the different levels of summarization requested in PROC SUMMARY, which depend on the number of variables in the CLASS statement. If you have one variable in the CLASS statement, variable *\_TYPE\_* will have 2 values: 0 and 1s, in which the 0 represents the grand total (count, average, or other requested statistics) and the 1s denote the totals for each value in the CLASS-variable. Similarly, if you have two variables in the CLASS statement, the 4 values of variable *\_TYPE\_* will range from 0, 1s, 2s and 3s. In this scenario, 0 would represent the overall total; the 1s would signify the totals for each value in the first variable in the CLASS statement; the 2s would stand for the totals for each value in the second variable in the CLASS statement; and finally, the 3s would represent the crosstabulation of values in the first and second variables in the CLASS statement.

The automatically generated variable *\_TYPE\_* is most beneficial when you specify two or more variables in the CLASS statement. Understanding each level of summarization and the value of *\_TYPE\_* that corresponds to that level enables you to control what level of analysis you want to show in your report (i.e., state, region, county). In this example, we are planning to present the state overall level (*\_TYPE\_* = 0), as well as the hospital level (*\_TYPE\_* = 1). Since there is no need to keep variable *\_TYPE\_*, use the DROP = option to suppress it from the outputted dataset *summary*.

The second automatically generated variable in PROC SUMMARY (*\_FREQ\_*) indicates the frequency or count of records for each level in the CLASS variable. In our example, it indicates total births at the state (*\_TYPE\_* = 0) and hospital (*\_TYPE\_* = 1) levels. You can use the RENAME = option to rename this variable to *Birth*<sup>1</sup>. The SUM = and MEAN = options are the requested statistics for the variables listed in the VAR statement ❹. They compute the counts and proportions, respectively. Note that the word “mean” may indicate statistical average but will work in this scenario since the data is categorical (1s or 0s). The AUTONAME option assigns the text strings “\_SUM” and “\_MEAN” to generated variable names containing computed frequencies and proportions. We could have used the AUTOLABEL option as well, but the flexibility of DEFINE statements in PROC REPORT (discussed later in this paper) gives you more control on the appearance of column headers.

#### STEP 2: ROUND AND SORT YOUR RESULTS IN A DATA STEP

```
DATA report1 (DROP = i);
  SET summary;
  ARRAY new (9) gestl_mean -- rec_err_mean; ❶
  DO i = 1 TO 9;
    new (i) = ROUND (new (i), 0.0001)*100;
```

<sup>1</sup> Or, you can leave the variable *\_FREQ\_* in dataset “summary” in step 1 and label it using DEFINE statement in step 5

```

END;
IF b hosp_ EQ 'Non-Hosp' THEN group = 1;
ELSE group = 0; *FORCES 'NON-HOSP' TO BE AT THE END ONCE SORTED; ❷

PROC SORT DATA = report1;
  BY group DESCENDING birth; ❸
RUN;

```

To further refine your summarized results, round the generated proportions and present them in a percentage form (i.e., 9.50% vs. 0.0950131234) in a second DATA step. The ARRAY statement ❶ above utilizes the ROUND function to round the computed proportions to 4 digits after the decimal. The multiplication by 100 will display the data in a more readable format. Since larger medical centers are of more interest to users of such report (they submit larger numbers of birth certificates), you want to maintain them at the beginning of the report. But, at the same time, you may want to keep the “non-hospital” group at the bottom. To do that, create the temporary variable *group* (drop later) and then use PROC SORT to sort the data by *group* first ❹ (variable will be sorted in ascending order by default, which keeps the string “non-hosp” after the numbers in a character variable), and then by the total number of births *birth*. Request a descending sort by placing the DESCENDING option before variable *birth*. This keeps hospitals with larger numbers of births at the top.

### STEP 3: COMPARE TO STANDARD

In this DATA step, we create 3 new variables: (1) *score85*, that represents the number of variables in which a hospital rate of reporting non-classifiable, unknown, or blank data fields ranges from 85% to equal to standard; (2) *score100*, that represents the number of variables in which a hospital rate of reporting poor data exceeds NCHS standards; and finally, *EBC*, which indicates whether a hospital uses Electronic Birth Certificate System (EBC) or not.

```

DATA report2 (DROP = group i);
  SET report1;
  ARRAY scorex (8) gest1_mean -- wtgain1_mean; ❶
  ARRAY standard (8) _TEMPORARY_ (1 17.78 5.64 5.03 2.24 2.24 2.99 7.79); ❷

  score85 = 0;
  score100 = 0;

  DO i = 1 TO 8;
    IF 0.85*standard (i) LE scorex (i) LE standard (i)
      THEN score85 = score85 + 1;
    ELSE IF scorex (i) GT standard (i)
      THEN score100 = score100 + 1;
  END;

  IF b hosp_ IN ('HOSPITAL-01', 'HOSPITAL-02', 'HOSPITAL-03',
               'HOSPITAL-04', 'HOSPITAL-05', 'HOSPITAL-06',
               'HOSPITAL-0X');
  THEN EBC = 'YES';
  ELSE EBC = 'NO'; ❸

  IF b hosp_ EQ '' THEN DO;
    b hosp_ = 'Total';
    EBC = 'N/A';
  END;

  IF birth LE 5 THEN DELETE;
RUN;

```

ARRAY *scorex* ❶ catalogs the percentages of the 8 selected DATA step variables while *standard* ❷ records the NCHS standard percentages. Note that we used the option *\_TEMPORARY\_* to create the list of standard percentages as temporary data elements. You can create new variables that analyses the characteristics of hospitals such as rural vs. urban, secondary vs. tertiary, etc. Here, we are using the IF-THEN/ELSE argument ❸ to classify hospitals as EBC vs. non-EBC hospitals. Delete hospitals reported less than 5 births during the period of evaluation.

**STEP 4: SET RANGES FOR COLOR-CODING USING PROC FORMAT**

The next FORMAT procedure assigns colors based on a decision rule<sup>2</sup>: orange if the percentage of a particular variable is 85% to equal to acceptable NCHS standards, and red if it exceeds NCHS standards.

```
PROC FORMAT;
  value visits    2.54-2.99 = orange  3.02-high = red other = black;
  value gest      0.85-1   = orange  1.01-high = red other = black;
  value menda    15.11-17.78 = orange 17.96-high = red other = black;
  value menmo    4.79-5.64 = orange  5.70-high = red other = black;
  value menyr    4.28-5.03 = orange  5.08-high = red other = black;
  value care     1.90-2.24 = orange  2.26-high = red other = black;
  value nammo    1.90-2.24 = orange  2.26-high = red other = black;
  value wtgain   6.62-7.79 = orange  7.87-high = red other = black;
RUN;
```

**STEP 5: APPLY THE FORMAT AND OTHER ODS STYLE DEFINITIONS WITHIN PROC REPORT**

```
ODS LISTING CLOSE;
ODS PDF FILE = "&path\SUGI31.pdf" NOTOC STYLE = SASWEB; ❶
PROC REPORT DATA = report2 NOWINDOWS
  STYLE (HEADER) = [FOREGROUND = YELLOW FONT_WEIGHT = BOLD FONT_SIZE = 8PT]; ❷
  COLUMN bhosp_ birth EBC score100 score85 rec_err_sum rec_err_mean
         gest1_sum gest1_mean mendal_sum mendal_mean menmol_sum  menmol_mean
         menyr1_sum menyr1_mean carel_sum carel_mean nammol_sum nammol_mean
         visits1_sum visits1_mean wtgain1_sum wtgain1_mean; ❸
```

Turn off the traditional LISTING destination and activate a PDF destination with a SASWEB STYLE ❶, which by default, generates a white FOREGROUND in column headers (Figure 1). Include the NOTOC option in the ODS PDF destination statement to suppress the table of contents in your PDF file (Figure 2). With a STYLE statement ❷, however, PROC REPORT gives you more control on column headers' foreground color, font weight, and font size (Figure 3). Like a VAR statement in PROC PRINT, COLUMN statement ❸ selects the arrangement of columns you want to present in your report.

Figure 1

Frequency	Percent	Cumulative Frequency	Cumulative Percent
12739	45.66	12739	45.66

Figure 2




Figure 3

Records with at Least 1 Error	Percent

```
DEFINE bhosp_/'Birth/Hospital';
DEFINE birth/'Birth/Count';
DEFINE score100/'Above/NCHS/Standard' STYLE(HEADER) = [FOREGROUND = RED];
DEFINE score85/'85% to/Equal/Standard' STYLE(HEADER) = [FOREGROUND = ORANGE];
```

THE DEFINE statements above describe how to use and display a report item in PROC REPORT, such as labeling your columns. With the STYLE (*ItemName*) = option, you can even control style elements (in ODS) for report items such as column headers. Here, we are requesting the column header for the variable *score100* be displayed in red, while column header for *score85* be displayed in orange (Figure 4). This serves as a "legend" of how to read the color-coded values in the report.

Figure 4

Birth Hospital	Birth Count	EBC	Above NCHS Standard	85% to Equal Standard
TOTAL	34290	N/A	1	0
HOSPITAL-01	2262	YES	0	0

Figure 5

Day Last Normal Menses	NCHS Standard -17.78%	Month Last Normal Menses	NCHS Standard -5.64%
1439	4.2	1399	4.08
0	0	0	0

Figure 6

Weight Gain	NCHS Standard -7.79%
3258	9.5

PROC STEP CONTINUED FROM PREVIOUS PAGE

```

DEFINE rec_err_sum/'Records/with at/Least/1 Error';
DEFINE rec_err_mean/'Percent';
DEFINE gestl_sum/'Clinical/Estimate/of/Gestation';
DEFINE gestl_mean/'NCHS/Standard/=1.00%'
  STYLE (COLUMN) = [ FOREGROUND = gest. FONT_WEIGHT = BOLD] ❶
  STYLE (HEADER) = [ FONT_STYLE = ITALIC];
DEFINE mendl_sum/'Day/Last/Normal/Menses';
DEFINE mendl_mean/'NCHS/Standard/=17.78%'
  STYLE (COLUMN) = [ FOREGROUND = menda. FONT_WEIGHT = BOLD] ❷
  STYLE (HEADER) = [ FONT_STYLE = ITALIC];
*Continue defining the rest of the variables the same way;

TITLE 'NATALITY NOT CLASSIFIABLE, UNKNOWN, AND BLANK DATA BY HOSPITAL: ARKANSAS
  2003'; ❸

RUN;
ODS ODF CLOSE;
ODS LISTING;
QUIT;

```

Recall that we did not use the AUTOLABEL option in PROC SUMMARY (step 1) because we wanted to take advantage of DEFINE statements in PROC REPORT. In this example, we are placing NCHS standards in the labels of the columns that contain the percentages. This works as a reference for comparison (Figure 5). Furthermore, this is where we apply the color-coding format ❶ and ❷, created in step 4, using the STYLE (COLUMN) = [FOREGROUND = *FORMAT-Name*] option (Figure 6). Add a title to your report using the TITLE statement ❸. Always place the RUN statement before ODS statements. Do not forget to turn off the ODS PDF destination and turn on the LISTING destination.

THE FINAL PRODUCT

The next 3 figures capture the generated report. The report is printed in a legal size paper with a landscape orientation.

NATALITY NOT CLASSIFIABLE, UNKNOWN, AND BLANK DATA BY HOSPITAL: ARKANSAS 2003																				01:18 Tuesday, July 6, 2004 1			
Hosp Hospital	Birth Count	EMC	Above NCHS Standard	50% to 90% Standard	Records with at Least 1 Error	Percent	Clinical Estimate of Gestation	NCHS Standard -1.00%	Day Last Normal Menses	NCHS Standard -17.78%	Month Last Normal Menses	NCHS Standard -6.64%	Year Last Normal Menses	NCHS Standard -6.63%	Month Prenatal Care Began	NCHS Standard -2.24%	Month Prenatal Care Began	NCHS Standard -2.24%	Prenatal Visits	NCHS Standard -2.80%	Weight Gain	NCHS Standard -7.79%	
TOTAL	34290	N/A	1	0	4942	14.41	152	0.44	1439	4.2	1399	4.08	1345	3.92	571	1.67	70	0.2	576	1.68	3258	9.5	
HOSPITAL-01	2262	YES	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
HOSPITAL-02	1977	YES	0	0	9	0.46	0	0	1	0.05	1	0.05	1	0.05	6	0.3	4	0.2	2	0.1	1	0.05	
HOSPITAL-03	1964	YES	1	0	1466	74.64	1	0.05	2	0.1	2	0.1	2	0.1	22	1.12	19	0.97	2	0.1	1460	74.34	
HOSPITAL-04	1783	YES	0	1	133	7.46	0	0	2	0.11	2	0.11	2	0.11	9	0.5	1	0.06	9	0.5	125	7.01	
HOSPITAL-05	1555	YES	3	0	427	27.46	4	0.26	40	2.57	39	2.51	39	2.51	70	4.5	0	0	71	4.57	357	22.96	
HOSPITAL-06	1482	YES	0	0	85	5.74	1	0.07	50	3.37	50	3.37	47	3.17	4	0.27	0	0	3	0.2	36	2.43	
HOSPITAL-07	1482	YES	0	0	26	1.75	5	0.34	6	0.4	6	0.4	5	0.34	12	0.81	10	0.67	2	0.13	13	0.88	
HOSPITAL-08	1428	YES	0	0	4	0.28	0	0	1	0.07	0	0	0	0	3	0.21	3	0.21	0	0	0	0	
HOSPITAL-09	1242	YES	0	0	92	7.41	1	0.08	33	2.66	33	2.66	32	2.58	18	1.45	1	0.08	18	1.45	46	3.7	
HOSPITAL-10	1228	YES	4	1	306	24.92	6	0.49	120	9.77	120	9.77	112	9.12	27	2.2	11	0.9	42	3.42	168	13.68	
HOSPITAL-11	1113	YES	0	0	31	2.79	0	0	8	0.72	8	0.72	8	0.72	2	0.18	0	0	1	0.09	23	2.07	
HOSPITAL-12	884	YES	6	0	411	46.49	5	0.57	352	39.82	352	39.82	331	37.44	26	2.94	0	0	75	8.48	72	8.14	
HOSPITAL-13	746	YES	1	1	59	7.91	1	0.13	40	5.36	40	5.36	40	5.36	5	0.67	0	0	11	1.47	8	1.07	
HOSPITAL-14	735	YES	6	0	183	24.9	25	3.4	61	8.3	61	8.3	61	8.3	62	8.44	0	0	31	4.22	73	9.93	
HOSPITAL-15	724	YES	5	0	135	18.65	10	1.38	76	10.5	76	10.5	76	10.5	37	5.11	3	0.41	51	7.04	43	5.94	
HOSPITAL-16	670	YES	3	1	202	30.15	1	0.15	90	13.43	90	13.43	79	11.79	13	1.94	0	0	11	1.64	148	22.09	
HOSPITAL-17	630	YES	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	
HOSPITAL-18	622	YES	0	0	38	6.11	1	0.16	0	0	0	0	0	0	4	0.64	0	0	3	0.48	34	5.47	
HOSPITAL-19	595	YES	0	0	39	6.55	2	0.34	25	4.2	25	4.2	24	4.03	1	0.17	1	0.17	3	0.5	11	1.85	
HOSPITAL-20	592	YES	0	0	14	2.36	3	0.51	4	0.68	4	0.68	4	0.68	7	1.18	1	0.17	0	0	1	0.17	
HOSPITAL-21	572	YES	6	0	177	30.94	31	5.42	69	12.06	69	12.06	69	12.06	33	5.77	0	0	30	5.24	92	16.08	
HOSPITAL-22	552	YES	4	0	72	13.04	2	0.36	63	11.41	63	11.41	63	11.41	51	9.24	1	0.18	23	4.17	21	3.8	
HOSPITAL-23	536	YES	1	0	34	6.34	3	0.56	17	3.17	17	3.17	17	3.17	27	5.04	0	0	10	1.87	5	0.93	
HOSPITAL-24	493	YES	5	0	146	29.61	1	0.2	66	13.39	66	13.39	66	13.39	31	6.29	0	0	69	14	47	9.53	
HOSPITAL-25	471	YES	0	0	6	1.27	1	0.21	3	0.64	3	0.64	3	0.64	2	0.42	0	0	2	0.42	3	0.64	
HOSPITAL-26	455	YES	2	0	74	16.26	1	0.22	20	4.4	20	4.4	18	3.96	7	1.54	0	0	19	4.18	37	8.13	
HOSPITAL-27	453	YES	0	0	7	1.55	0	0	0	0	0	0	0	0	0	0	0	0	0	0	7	1.55	
HOSPITAL-28	440	YES	1	0	36	8.18	8	1.82	16	3.64	16	3.64	16	3.64	3	0.68	0	0	7	1.59	14	3.18	
HOSPITAL-29	411	YES	0	2	26	6.33	0	0	23	5.6	20	4.87	20	4.87	2	0.49	0	0	0	0	3	0.73	



Birth Hospital	Birth Count	EBC	Above NCHS Standard	88% to Equal Standard	Records with at Least 1 Error	Percent	Clinical Estimate of Gestation	NCHS Standard =1.00%	Day Last Normal Menses	NCHS Standard =17.78%	Month Last Normal Menses	NCHS Standard =5.64%
TOTAL	34290	N/A	1	0	4942	14.41	152	0.44	1439	4.2	1399	4.08
HOSPITAL-01	2262	YES	0	0	0	0	0	0	0	0	0	0
HOSPITAL-02	1977	YES	0	0	9	0.46	0	0	1	0.05	1	0.05
HOSPITAL-03	1964	YES	1	0	1466	74.64	1	0.05	2	0.1	2	0.1
HOSPITAL-04	1783	YES	0	1	133	7.46	0	0	2	0.11	2	0.11
HOSPITAL-05	1555	YES	3	0	427	27.46	4	0.26	40	2.57	39	2.51
HOSPITAL-06	1482	YES	0	0	85	5.74	1	0.07	50	3.37	50	3.37
HOSPITAL-07	1482	YES	0	0	26	1.75	5	0.34	6	0.4	6	0.4
HOSPITAL-08	1428	YES	0	0	4	0.28	0	0	1	0.07	0	0
HOSPITAL-09	1242	YES	0	0	92	7.41	1	0.08	33	2.66	33	2.66
HOSPITAL-10	1228	YES	4	1	306	24.92	6	0.49	120	9.77	120	9.77
HOSPITAL-11	1113	YES	0	0	31	2.79	0	0	8	0.72	8	0.72
HOSPITAL-12	884	YES	6	0	411	46.49	5	0.57	352	39.82	352	39.82
HOSPITAL-13	746	YES	1	1	59	7.91	1	0.13	40	5.36	40	5.36
HOSPITAL-14	735	YES	6	0	183	24.9	25	3.4	61	8.3	61	8.3
HOSPITAL-15	724	YES	5	0	135	18.65	10	1.38	76	10.5	76	10.5
HOSPITAL-16	670	YES	3	1	202	30.15	1	0.15	90	13.43	90	13.43

Day Last Normal Menses	NCHS Standard =17.78%	Month Last Normal Menses	NCHS Standard =5.64%	Year Last Normal Menses	NCHS Standard =5.03%	Month Prenatal Care Began	NCHS Standard =2.24%	Named Month Prenatal Care Began	NCHS Standard =2.24%	Prenatal Visits	NCHS Standard =2.99%	Weight Gain	NCHS Standard =7.79%
1439	4.2	1399	4.08	1345	3.92	571	1.67	70	0.2	576	1.68	3258	9.5
0	0	0	0	0	0	0	0	0	0	0	0	0	0
1	0.05	1	0.05	1	0.05	6	0.3	4	0.2	2	0.1	1	0.05
2	0.1	2	0.1	2	0.1	22	1.12	19	0.97	2	0.1	1460	74.34
2	0.11	2	0.11	2	0.11	9	0.5	1	0.06	9	0.5	125	7.01
40	2.57	39	2.51	39	2.51	70	4.5	0	0	71	4.57	357	22.96
50	3.37	50	3.37	47	3.17	4	0.27	0	0	3	0.2	36	2.43
6	0.4	6	0.4	5	0.34	12	0.81	10	0.67	2	0.13	13	0.88
1	0.07	0	0	0	0	3	0.21	3	0.21	0	0	0	0
33	2.66	33	2.66	32	2.58	18	1.45	1	0.08	18	1.45	46	3.7
120	9.77	120	9.77	112	9.12	27	2.2	11	0.9	42	3.42	168	13.68
8	0.72	8	0.72	8	0.72	2	0.18	0	0	1	0.09	23	2.07
352	39.82	352	39.82	331	37.44	26	2.94	0	0	75	8.48	72	8.14
40	5.36	40	5.36	40	5.36	5	0.67	0	0	11	1.47	8	1.07
61	8.3	61	8.3	61	8.3	62	8.44	0	0	31	4.22	73	9.93
76	10.5	76	10.5	76	10.5	37	5.11	3	0.41	51	7.04	43	5.94
90	13.43	90	13.43	79	11.79	13	1.94	0	0	11	1.64	148	22.09

## CONCLUSION

In MCH and other reproductive health research and services, edited and complete birth data produces reliable and significant results. The Center for Health Statistics at the Arkansas Division of Health is highly rated from the National Center for Health Statistics (NCHS) regarding the accuracy, timeliness, and completeness of their data. To carry out quality MCH research for Arkansas citizens and to maintain these high ratings, the center evaluates the performance of birth hospitals concerning submitting invalid data in birth certificates. To accomplish that, they are putting SAS and its Output Delivery System (ODS) to use.

The introduction of ODS motivated us to move from the multi-step SAS-Excel process of generating color-coded reports to an all-in-one process using PROC REPORT. The REPORT procedure combines features of the PRINT, MEANS, and TABULATE procedures with features of the DATA step in a single report-writing tool<sup>3</sup>. In this project, we are not using all of these features, nevertheless; the incorporation of color-coding formats into PROC REPORT is of exceptional utility for users of periodical reports in which certain levels of performance need to be highlighted.

The default features provided by ODS in the different versions of SAS are convenient and can fit any report. To create customized reports, however, you may need to alter or modify defaults using the flexibility of ODS style definitions. PROC REPORT is an excellent venue to use these style definitions.



## REFERENCES

1. Centers for Disease Control and Prevention (CDC). National center for Health Statistics (NCHS). <http://www.cdc.gov/nchs/nvss.htm>.
2. Karp, Andrew. "ODS 101: Essential Concepts of the SAS® Output Delivery System", Sierra Information Services, Inc. Sonoma, California. <http://www.SierralInformation.com>.
3. SAS Institute Inc. 2002, SAS 9 Output Delivery System: User's Guide. Cary, NC

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact us at:

Thaer Baroud, MHSA, MA  
Senior Epidemiologist  
Arkansas Department of Health and  
Human Services-Division of Health  
Center for Public Health Practice  
Epidemiology Branch  
P. O. Box 1437, Slot # H 32  
Little Rock, AR 72203  
Phone: (501) 280-4812  
Email: [tbaroud@healthyarkansas.com](mailto:tbaroud@healthyarkansas.com)

John Senner, PhD  
Branch Chief, Senior Research Analyst  
Arkansas Department of Health and  
Human Services-Division of Health  
Center for Public Health Practice  
Health Statistics Branch  
P. O. Box 1437, Slot # H 19  
Little Rock, AR 72203  
Phone: (501) 661-2497  
Email: [jsenner@healthyarkansas.com](mailto:jsenner@healthyarkansas.com)

Paul Johnson, MSPH  
Senior Research Analyst  
Arkansas Department of Health and  
Human Services-Division of Health  
Center for Public Health Practice  
Health Statistics Branch  
P. O. Box 1437, Slot # H 19  
Little Rock, AR 72203  
Phone: (501) 661-2861  
Email: [pwjohnson@healthyarkansas.com](mailto:pwjohnson@healthyarkansas.com)

*SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® Indicates USA registration.*

*Other brand and product names are registered trademarks or trademarks of their respective companies.*